Appendix: Non-Dimensionalizing (Scaling, or Normalizing)

Units and Numbers

Many readers will find the following discussion of units and non-dimensional numbers to be banal, or too many words for a topic which everyone understands intuitively. Perhaps, they are right. However, I believe that what follows produces a best-practice technique to organize thoughts about the nature of a model and the manner in which results are communicated most efficiently and naturally.

I also believe that non-dimensionalizing (also known as scaling or normalizing) should be *a preliminary* step to developing any model.

1. Non-dimensionalizing helps the modeler decide which are the relevant variables and how they might be related.

2. It provides a technique to do dimensional analysis[1].

3. It reduces the number of extraneous symbols that appear in calculations that are the origin of silly mistakes.

4. It eliminates the possibility of reporting nonsense such as the logarithm of a kilogram[2].

It is better and correct to report "log(mass) in SI units" which is another way of saying the mass is the *ratio* of the reported/predicted mass to a unit kilogram. The discussion that follows a proposition that *it is usually the best practice* to report results as a dimensionless ratio of the reported quantity *to a unit that is relevant to the physics of the problem.* For example, "log(μ) where μ is the ratio of the mass to the mass of the sun" when the model is about planets—and "... where μ is the ratio of the mass to that of a neutron" when the model is about atoms.

If dimensionless ratio is order unity, then so much the better: "The prediction is that the star has 12.5 solar masses," or "We have measured band gap and it is 1.5 times larger than that of pure silicon at room temperature." These numbers are easier to interpret than $m_{\text{star}} = 2.5 \times 10^{31}$ kilograms and $E_{\text{gap}} = 1.8 \times 10^{-19}$ joules.

Of course, sometimes it is *required* that a result should be reported in a unit. If so, it is easy to turn a dimensionless quantity into one that has units by multiplication.

A Whimsical Metaphor

Suppose that we are looking for evidence of extraterrestrial intelligence—or wish to broadcast that intelligence exists on earth—though pulses of electromagnetic radiation. How would such a thing be accomplished?

We could broadcast numbers with fixed short pulses and pauses: 3 pulses, pause; 1 pulse, pause; 4 pulses, pause; 1 pulse, pause... This goes on for a while, and there are are no pulses larger than 9 (we might use two consecutive pauses as a zero–the absence of a pulse). After it becomes sufficiently unlikely that there are going to be any pulse sequences greater than nine, a hypothesis can be formed, "hmmm (or an alien equivalent thereof), this is probably something in base 10." In fact the ratio of pauses to pulses is exactly 10. If these pulses could be translated to a more convenient base, say 7, it would begin to look like the digits of π in base 7. In fact they listen long enough to establish that they can predict the next set of pulses with extremely high certainty.

New hypotheses can be formed. "This is either a very unusual physical phenomenon, or evidence that there is probably another intelligence in the direction of these pulses." After many pulses have

[1] Dimensional analysis in a nutshell: how one measureable quantity will depend on another (e.g, if the size of a sample is doubled and everything else remains the same, then the concentration profile will have the same characteristics at a time that is four times longer)

[2] Nevertheless, quantities like "log(mass) are often reported which is–I believe–a deplorable practice

been sent, there is a long break, and then new sequence is transmitted—2 pulses; 7 pulses; 1 pulse; 8 pulses—indicating the digits in base 10 of e. The hypothesis of other intelligent life becomes more and more plausible.

Additional hypotheses would follow such as: a knowledge of geometry (vis-a-vis π); a knowledge of calculus (e); they have developed enough technology to generate and control electromagnetic pulses; their standard unit of time is about the length of a pause; they probably learned how to count using 10 things ("What an odd being it would be to have an even number of appendages. If they had the same number on both sides, it would be too difficult to know left-handed from right-handed: they probably have three on one side and seven on the other, because seven is so *intuitively natural*")

The fictional communication above is plausible because *pure numbers* are being transmitted; mutual agreement of a standard unit is unnecessary.

Before reading on, reflect on this question for a minute or two: "How would we convey that we might know something about the special theory of relativity; for example, that the speed of light in a vacuum is always the same independent of the inertial reference frame."

It wouldn't be sensible to broadcast 2-9-9-7-9-2-5... (i,e, $c \approx 2.997925 \times 10^8$ meters/second). There is know way of indicating a unit of distance and a unit of time.

However, "important" quantities like the speed of light do appear in dimensionless number that, as far as we know, are the same everywhere in the universe. For example, the fine structure constant $\alpha = e^2/(\epsilon_o \hbar c)$ (i.e., the square charge of an electron divided by the product of [the permittivity of free space] [Planck's constant divided by 2π] [the speed of light]) appears frequently in atomic physics, The number 137 is frequently a favorite among physical scientists because $1/\alpha$ is approximately 137 ($\alpha = 1/137.036$)

So, the sequence 1-3-7-0-3-... is transmitted (presumably with as many digits as we have established by experiment). They form the hypothesis that we know something about important physical constants like c and $\hbar$—and perhaps the technical quality of our experimental apparatus. After this we might send our value of c in units of $e^2/(\alpha \epsilon_o \hbar)$ and perhaps they could infer the ratio (meters/second) that we use as a standard for speed.

There are many holes in the logic of this metaphor, but the significance is clear: the possibility of a misunderstanding or error is reduced when physical quantities are non-dimensionalized.

Dimensions and Physics

There can be no physical consequence of the choice of units. The half-life of carbon-14 is the same regardless of whether that data is reported in seconds, years, multiples of the world record for the 100 meter dash, or the time it takes a photon to travel one Bohr radius. Of course, the number looks different, but the physical quantity is the same. This creates a bit of overhead, one has to keep track of the units and make sure they are consistent. Experience shows that misunderstanding about units can lead to unfortunate errors.

However, the ratio of half-life of Uranium-236 to that of Carbon-14 is a fixed quantity—*independent of which units are used*—it is always 0.008. This ratio has no units: it is just a number and nothing else except what it represents. In many cases, where the two half-lives are pertinent to the physical problem that is being modeled, this ratio is more illustrative than the individual half-lives. In the case where the individual numbers are important, many scientists know that the half-life of Carbon-14 is about 5700 years: U-235 is easy to calculate from the ratio.

Other instances where using numbers-without-units are plotting and manipulating data. For example, Arrhenius plots have $\log(c)$ as the y-axis and $1/T$ as the x-axis. (Figure 1) However, the concentration, c, is sometimes reports in units of 1/meter^3, 1/cm^3, or some other inverse-volume measure. What does it mean to take the logarithm of 1/meter^3? The log doesn't make sense: note that the expansion of $\log(x)$ around $x = 1$ is $(x-1) - \frac{1}{2}(x-1)^2 + \frac{1}{3}(x-1)^3 + O\left((x-1)^4\right)$: all powers of x appear and wouldn't make sense if x had a unit associated with it.

One way to get around this is to make the units non-dimensional. For example, define new dimensionless unit:

$$\tau \equiv T/T_{\text{melting}} \quad \text{and} \quad \bar{c} \equiv c(T)/c(T_{\text{melting}}) \tag{1}$$



Figure 1: Examples of plotting kinetic data with the Arrhenius relation where $T_{\text{melt}} = 873$K and $c(T_{\text{melt}}) = 8.7 \times 10^{-17}/\text{meters}^3$. Each represents the same data: the first with SI units, the second with normalized quantities, the third with "self-explanatory" graphics. Which is better? The slopes are meaningful too; how should they be reported?

Every model should produce a measurable result than be used to compare an experiment or to make a prediction. The types of measurement can be placed into two categories:

Dimensional: Measurements/predictions that depend on *unit dimension* that is agreed upon or understood by all interested parties. For example, *average miles per gallon* for gas mileage and for the coefficient of linear thermal expansion, *change-in-length/(reference-length×temperature)* which is typically reported as 2×10^{-6}/Kelvin for hard materials, but may be more "intuitive" if reported as 2 microns/(meter×Kelvin).

Non-Dimensional Measurements or predictions that have no units attached to them, such as the ratio of the average diameter of a circular object to its perimeter "numbers" (or better yet the average diameter of a circular object to its perimeter divided by π: a measure of "nearness" to a circle).

Measurements or predictions that are scaled by measured physical constants, such as the velocity of an airplane divided by the velocity of sound in air at STP; the measured density of blood at different temperature divided by the density of pure water at STP; the velocity of a neutron divided by $e^2/(\epsilon_o \hbar)$.

Measurements or predictions that are scaled by quantities that pertain to a particular model. For example, the height of a cylinder l can be normalized by introducing a non-dimensional height $\lambda = l/R$ where R is the cylinder radius; the volume of the cylinder is $\lambda \pi R^3$ (all the length units appear as R). Another is sample is the force on a spring: $F = k(x - x_o)$. A non-dimensional length $\epsilon \equiv (x - x_o)/x_o$ and a non-dimensional force $\phi = F/(kx_o)$, to produce a non-dimensional form of Hooke's law: $\phi = \epsilon$. In other words, the force versus displacement plot for *every linear spring* looks the same if one plots $F/(kx_o)$ on the y-axis and $(x - x_o)/x_o$ on the x-axis.

Example: Scaling the Lennard-Jones Potential

A simple and useful model for the interaction between two atoms is the Lennard-Jones potential:

$$U(r) = \frac{a}{r^{12}} + \frac{b}{r^6} \tag{2}$$

and is sometimes called the 6–12 potential. The rationale for the potential is that there is a short-range (a/r^{12}) repulsive term that models the overlap of the electronic orbitals of two atoms; the long-range ($-b/r^6$) attractive term that can be derived from two fluctuating dipoles[3]. The reference zero-energy in this model is chosen at $r = \infty$. With this rationale, the potential $U(r)$ in Eq. 2 should be attractive at large r (i.e., positive slope), repulsive at small r (negative slope) and have a minimum (zero slope) at the equilibrium separation r_{min}. So, given the function in Eq. 2 in parameters a b, and a quick way to plot and verify that the function has this behavior, what do beginners (and, sadly, many experienced scientists and engineers) do? They guess (Fig. 2).



Figure 2: Hit and miss method to visualize physical behavior of a model: given parameters a and b, how does one find the right ballpark for picking these parameters? It is bad enough to guess for only two parameters; it becomes much worse of their are many model parameters. *Even worse, a and b have energy-length units; what does it mean to pick a number for their values? Once picked, what are the meaning of the units for the energy and distance in the plot?*

There is a methodical—and in my opinion superior—method: *non-dimensionalizing*.

In Eq. 2 there are two characteristic parameters: the equilibrium separation and the depth of the well. There are two meaningful physical quantities; there are two opaque model parameters. The method follows from solving two equations for two unknowns:

$$\frac{d}{dr}\left(\frac{a}{r^6} + \frac{b}{r^{12}}\right)\Bigg|_{r=r_{\text{min}}} \equiv 0 \qquad \rightarrow \qquad r_{\text{min}}^6 = -\frac{2b}{a} \tag{3}$$

$$\left(\frac{a}{r_{\text{min}}^6} + \frac{b}{r_{\text{min}}^{12}}\right) \equiv -E_{\text{min}} \tag{4}$$

[3]Sometimes called London dispersion forces

where E_{min} is chosen as a positive term so that the minimum is located at a negative value, $-E_{\text{min}}$, relative to the zero at $r = \infty$. Solving for a and b in terms of r_{min} and E_{min} and inserting into Eq. 2:

$$\begin{aligned} a &= -2E_{\text{min}} r_{\text{min}}^6 \\ b &= E_{\text{min}} r_{\text{min}}^{12} \end{aligned} \qquad \rightarrow \qquad \frac{U(r)}{E_{\text{min}}} = \left(\frac{r_{\text{min}}^{12}}{r^{12}} - \frac{2r_{\text{min}}^6}{r^6} \right) \tag{5}$$

Both sides of the above equation are non-dimensional. This can be taken one step futher by introducing a characteristic non-dimensional energy $\rho \equiv r/r_{\text{min}}$ and $\upsilon(\rho) \equiv U(r)/E_{\text{min}}$:

$$\upsilon(\rho) = \frac{1}{\rho^{12}} - \frac{2}{\rho^6} \tag{6}$$

All the physics of the Lennard-Jones potential can be found in Eq. 5 or Eq. 6. In hindsight, the form of the dimensionless equation could have been guessed without solving: the function had to be $U(r_{\text{min}}) = -E_{\text{min}}$ and the powers of the length terms had to be six and twelve.

It is now a simple matter to derive other physical quantities—expanding the energy about its minimum:

$$\upsilon(\rho) - \upsilon(\rho = 1) = \frac{k}{2}(\Delta\rho)^2 + \frac{\alpha}{6}(\Delta\rho)^3 = \frac{72}{2}(\Delta\rho)^2 - \frac{1512}{6}(\Delta\rho)^3 \tag{7}$$

The cofactor $k = 72$ is the non-dimensional spring-constant for this Lennard-Jones system; the dimensional spring-constant is $72 \times E_{\text{min}}/r_{\text{min}}^2$. The characteristic atomic oscillation frequency must be $\overline{\omega} = \sqrt{k/(M_{\text{atom}}/M_{\text{proton}})} = \sqrt{72/\mu}$. Thus the dimensinal characteristic oscillation frequency is

$$\omega \approx 6\sqrt{2/Z} \times \sqrt{\frac{E_{\text{min}}}{M_{\text{proton}}}} \frac{1}{r_{\text{min}}} \tag{8}$$

and this provides a new dimensionless quantity, a characteristic time: $\tau =$ The latent heat of evaporation for argon is about 1/10 electron volt, we can take that as an approximation to $E_{textmin}$, r_{min} is probably around 10 angstroms, a proton has a mass of about 10^{-27} kilograms; so the characteristic frequency for bound argon is about 10^{12}/second. For an atom with four times the mass, the frequency would be about half.

Scaling and non-dimensionalizing frequently permits a rapid comparision similar physical systems, but with differing parameters. I believe putting numbers in from the very beginning is much less instructive—and much more prone to errors due to unit matching on both sides of an equation.

There is also a physical interpretation of the cubic term $(\Delta\rho)^3$ in Eq. 7. This cubic term has a negative coefficient; therefore the total energy is lowered if the atom spends a bit more time at $\Delta\rho > 0$ (i.e., expanded). This is a useful way to interpret the physics of the coefficient of thermal expansion.

Here is an example of how to do this Lennard-Jones scaling with Mathematica.

Example Scaling for Lennard-Jones Potential

Example of Non-Dimensionalizing: The Lennard-Jones Potential

Download notebooks, pdf(color), pdf(bw), or html from http://pruffle.mit.edu/3.016-2009.

In this example, a suggested sequence of MATHEMATICA® operations to non-dimensionalize a potential is presented.

Example of Non-Dimensionalizing the Lennard-Jones Potential

```
LJPotential = a/r^6 + b/r^12                                              1
```

Find the solutions for extrema of the potential in terms of r by solving for the zeros of the derivative *A*

```
rminsol = Simplify[Solve[D[LJPotential, r] == 0, r]]                      2
```

This result may look a bit confusing at first, but they are all the same solution. An alternative way to observe this is shown at the end of this notebook.

Define Eminsol as the value at the minimum *B*

```
Eminsol = LJPotential /. rminsol                                          3
```

Solve for a and b in terms of the physical parameters Emin and rmin (using the second of the six solutions). Using Emin as positive, then -Emin is used for the minimum. *C*

```
absols = Solve[{rmin == r /. rminsol[[2]], -Emin == Eminsol[[2]]}, {a, b}]    4
```

This illustrated that all the solutions are the same. *D*

```
Solve[{rmin == r /. rminsol[[3]], -Emin == Eminsol[[3]]}, {a, b}]          5
```

Non-dimensionalize the potential by dividing by an energy. Here we want the potential to be -1 at the minimum, so we divide by the potential by Emin *E*

```
LJScaled = Expand[LJPotential/Emin /. absols][[1]]                        6
```

The right hand side is non-dimensional, introduce a non-dimensional variable ρ *F*

```
LJDimensionless = LJScaled /. r → ρ rmin                                  7
```

```
Series[LJScaled, {r, rmin, 3}]
Series[LJDimensionless, {ρ, 1, 3}]                                        8
```

```
Plot[LJDimensionless, {ρ, .5, 3}, PlotLabel → "Non-Dimensionalized Lennard-Jones Potential", AxesLabel → {"r/r_min", "Potential/E_min"}]    9
```

1 Define the Lennard-Jones potential in terms of a and b.

2 Solve for the mininum; name the rule for the solution. The result may be a bit odd, but the value of r^6 is the *same* for all the solutions. Because the potential depends only on r^6 and $r^{2\times 6}$, this makes sense.

3 Replace the solution to find the value of the minima.

4–5 Introduce the two physical variables through equalities.

6 Introduce the dimensionless energy by dividing through by the characteristic energy.

7 Introduce the dimensionless length using a rule.

8–9 Examples of looking at the expansion near the stable solution and a plot that is easier to understand in terms of physical variables.

It would be possible to define all quantities up from (e.g., $g = -9.9 meter/second^2$) and then type expressions such as `x = xo + v t + g t2̂/2` to make sure units are consistent. But, I think this is not such a good idea because it doesn't take advantage of Mathematica's symbolic capabilities.

Nevertheless, there are ways to automatically keep track of units. Here is an example using the Units package.

Example Using Units with the Units Package

Example of Using Units in Calculationsl

Download notebooks, pdf(color), pdf(bw), or html from http://pruffle.mit.edu/3.016-2009.

In this example, Mathematica's Units package is used to keep track of–and mix–units in a computation.

Examples of Using the Units Package

Loading the Package (A)

```
Needs["Units`"]
```
(1)

Simple Example of an Unusual Conversion of Units (B)

```
Convert[24 Mile/Gallon, 1/Acre]
```
(2)

Example of Mixing Units in a Simple Physics Problem (C)

Motion of an object that obeys the ODE $\frac{d^2 x}{dt^2}$=-g

```
x = xo + vo t - g t^2/2
velocity = D[x, t]
```
(3)

Define a particular instance of velocity by giving initial conditions and constants in different units

```
ParticularVelocity = velocity /. {g → 32 Foot/(Second)^2, vo -> 1024 Milli Meter/Minute, xo -> 1 Yard, t → 1.5 Minute}
```
(4)

The velocity remains in mixed units, but is easily converted to a consistent unit

```
Convert[ParticularVelocity, Mile/Day]
```
(5)

And additional example for position

```
conditions = {g → 9.8 Meter/(Second)^2, vo → 10 000 League/Fortnight, xo → 0, t → 1 Second}

Convert[x /. conditions, Meter]
```
(6)

1 Loading in the Units package.

2 Miles per gallon has units of inverse area and so does 1/acre: here is an example of how to find the conversion factor.

3 Define an example model for the position of an object in free-fall through a vacuum. Assign the correct expression to the velocity by using a derivative of the position.

4 Use a Replace with a set of rules to find a particular velocity. In this example, the units for the variables are chosen inconsistently. It is an artificial example, but will turn out fine. The result will have strange mixed-up units as well.

5 The `Convert` function in the `Units` package allows the expression to reported in chosen units.

6 Here is an additional example for the position.

Example: Scaling a Partial Differential Equation (The Diffusion Equation)

Sometimes, it can be very instructive to non-dimensionlize variables within differential equations. For an example, consider the diffusion equation which is a model for the time-evolution of the concentration as a function of its position, or $c(x, t)$. The concentration is a function of *where* and *when* (For example, the concentration will have the value 1 mole/cubic meter at 0.01 meters below the surface after 12 minutes; or, at 10 minutes, the spatial variation of the concentration is given by a spatial function $c(x, t = 12\text{min})$; or, considering the point at $x = 0.01$meter, the time variation of the concentration will be by a spatial function $c(x = 0.01\text{meter}, t))$. The diffusion equation has the form:

$$\frac{\partial c(x,t)}{\partial t} = D\frac{\partial^2 c(x,t)}{\partial x^2} \tag{9}$$

D is known as the diffusion coefficient—it is a material quantity that depends on what species is flowing (diffusing) and through what medium—what are its units? Because the units must match on both sides of Eq. 9, D must have the same units as x^2/t (i.e., length^2/time. The concentration $c(x,t)$ is the same on both sides of the diffusion equation 9—therefore it doesn't matter if the *value* of concentration is reported in somethings/(cubic meter) or in somethings/(cubic foot), as long as it is consistent. However, the units of position inside $c(x,t)$ must be the same as that of x in the ∂x^2.

It is conventional to introduce a characteristic "diffusion length L_D and t_D," where $L_D = \sqrt{Dt_D}$" which is a rough approcimation to how far a species would travel in time t_D. In any case, the quantity

$$\eta \equiv \frac{x}{\sqrt{Dt}} \tag{10}$$

is a dimensionless *variable*.

As shown below, introducing η into the Eq. 9 is a clever trick. First, for the left-hand-side of Eq. 9:

$$\frac{\partial c}{\partial t} \to \frac{\partial \eta}{\partial t}\frac{\partial c}{\partial \eta} = \frac{-x}{2D^{1/2}t^{3/2}}\frac{\partial c}{\partial \eta} \tag{11}$$

where the definition in Eq. 10 is used in the last step above. For the right-hand-side of Eq. 9:

$$\begin{aligned} \frac{\partial c}{\partial x} &\to \frac{\partial \eta}{\partial x}\frac{\partial c}{\partial \eta} \\ \frac{\partial^2 c}{\partial x^2} &\to \frac{\partial^2 \eta}{\partial x^2}\frac{\partial c}{\partial \eta} + \left(\frac{\partial c}{\partial \eta}\right)^2 \frac{\partial^2 c}{\partial \eta^2} = \left(\frac{\partial c}{\partial \eta}\right)^2 \frac{\partial^2 c}{\partial \eta^2} = \frac{1}{Dt}\frac{\partial^2 c}{\partial \eta^2} \end{aligned} \tag{12}$$

The diffusion equation becomes:

$$\frac{\partial c}{\partial t} = D\frac{\partial^2 c}{\partial x^2} \to \frac{dc}{d\eta} = \frac{-2}{\eta}\frac{d^2 c}{d\eta^2} \tag{13}$$

where the $\partial \to d$ because η is the only variable in the differential relation.

The scaled equation 13 is easy to solve, let $q \equiv dc/d\eta$ then

$$\begin{aligned} \frac{dq}{d\eta} &= -\frac{\eta q}{2} \\ \frac{dq}{q} &= \frac{-\eta}{2}d\eta \to \int \frac{dq}{q} = \int \frac{-\eta}{2}d\eta \\ \text{integrating once} \qquad \log\frac{q}{C_1} &= -\eta^2 \to q \equiv \frac{dc}{d\eta} = C_1 \exp(-\eta^2) \\ \text{integrating again} \qquad c(\eta) &= C_2 + \sqrt{\pi}C_1 \text{erf}\left(\frac{\eta}{2}\right) \end{aligned} \tag{14}$$

$$c(x,t) = c_0 + (c_\infty - c_0)\text{erf}\left(\frac{x}{\sqrt{4Dt}}\right)$$

Here is an example of how to do the above derivation in Mathematica:

Example Using Non-Dimensional Variables in the Diffusion Equation

Example of Non-Dimensionalizing a Partial Differential Equation

Download notebooks, pdf(color), pdf(bw), or html from http://pruffle.mit.edu/3.016-2009.

In this example, the diffusion one dimensional equation $\partial C/\partial t = D\partial^2 c/\partial x^2$ is converted to an ODE by introducing the non-dimensional quantity: $\eta \equiv x/\sqrt{\eta}\sqrt{Dt}$.

1 **Scaling the diffusion equation:** $\frac{\partial c}{\partial t} = D\,\frac{\partial^2 c}{\partial x^2}$

Try c[x,t] ⟶c[η[x,t]] and investigate the forms of the left- and right-hand-sides of the diffusion equation

left-hand-side, introduce a temporary variable for η, the rule x → η Sqrt[Diffusivity t] will be used later

```
etatemp = x/Sqrt[Diffusivity t]    1
```

Instead of c[x,t], try c[eta[x,t]]: left-hand-side = $\frac{\partial c}{\partial t}$

```
lhs = D[c[etatemp], t]    2
```

right-hand-side D $\frac{\partial^2 c}{\partial x^2}$

```
rhs = Diffusivity D[c[etatemp], {x, 2}]    3
```

Plug the lhs and rhs into the diffusion equation, and use the rule x → η Sqrt[Diffusivity t], this turns the partial differential equation into an ordinary differential equation.

```
DE = FullSimplify[(lhs == rhs) /. x → η Sqrt[ Diffusivity t], Assumptions → {t > 0, Diffusivity > 0}]    4
```

```
η c'[η] + 2 c''[η] == 0
```

This is one form of the solution with arbitrary constants

```
DSolve[DE, c[η], η]    5
```

This is one form of the solution with meaningful constants

```
DSolve[{DE, c[0] == czero, c[∞] == cinf}, c[η], η] // Simplify    6
```

$$\{\{c[\eta] \to czero + (cinf - czero)\,Erf[\tfrac{\eta}{2}]\}\}$$

1 Introduce a symbol for the dimensionless unit η. The symbol η will be used later, so here another placeholder is used.

2 Define the left-hand-side of the diffusion equation $\partial c/\partial t$.

3 Define the right-hand-side of the diffusion equation $D\partial^2 c/\partial t^2$.

4 Using a logical equals, a rule to introduce the symbol η, and a Simplify, the diffusion equation becomes a second order ODE in the variable η.

5 Here, DSolve is used to find the general solution.

6 Here is the solution with boundary conditions at $\eta = 0$ and $\eta = \infty$.

Why wouldn't one use this non-dimensionlizing trick with the diffusion equation *all the time?*. The answer is that the boundary conditions must be invariant to the scaling $\eta = x/\sqrt{Dt}$. This kind of scaling usually holds when the domain is infinite (i.e., half-space), but would be very unusual for finite boundary conditions.